


This article was downloaded by: [Zhang, Zhiyong]

On: 11 October 2008

Access details: Access Details: [subscription number 903562060]

Publisher Psychology Press

Informa Ltd Registered in England and Wales Registered Number: 1072954 Registered office: Mortimer House, 37-41 Mortimer Street, London W1T 3JH, UK



Structural Equation Modeling: A Multidisciplinary Journal

Publication details, including instructions for authors and subscription information:

<http://www.informaworld.com/smpp/title-content=t775653699>

A SAS Interface for Bayesian Analysis With WinBUGS

Zhiyong Zhang ^a; John J. McArdle ^b; Lijuan Wang ^a; Fumiaki Hamagami ^c

^a University of Notre Dame, ^b University of Southern California, ^c University of Virginia,

Online Publication Date: 01 October 2008

To cite this Article Zhang, Zhiyong, McArdle, John J., Wang, Lijuan and Hamagami, Fumiaki(2008)'A SAS Interface for Bayesian Analysis With WinBUGS',Structural Equation Modeling: A Multidisciplinary Journal,15:4,705 — 728

To link to this Article: DOI: 10.1080/10705510802339106

URL: <http://dx.doi.org/10.1080/10705510802339106>

PLEASE SCROLL DOWN FOR ARTICLE

Full terms and conditions of use: <http://www.informaworld.com/terms-and-conditions-of-access.pdf>

This article may be used for research, teaching and private study purposes. Any substantial or systematic reproduction, re-distribution, re-selling, loan or sub-licensing, systematic supply or distribution in any form to anyone is expressly forbidden.

The publisher does not give any warranty express or implied or make any representation that the contents will be complete or accurate or up to date. The accuracy of any instructions, formulae and drug doses should be independently verified with primary sources. The publisher shall not be liable for any loss, actions, claims, proceedings, demand or costs or damages whatsoever or howsoever caused arising directly or indirectly in connection with or arising out of the use of this material.

TEACHER'S CORNER

A SAS Interface for Bayesian Analysis With WinBUGS

Zhiyong Zhang

University of Notre Dame

John J. McArdle

University of Southern California

Lijuan Wang

University of Notre Dame

Fumiaki Hamagami

University of Virginia

Bayesian methods are becoming very popular despite some practical difficulties in implementation. To assist in the practical application of Bayesian methods, we show how to implement Bayesian analysis with WinBUGS as part of a standard set of SAS routines. This implementation procedure is first illustrated by fitting a multiple regression model and then a linear growth curve model. A third example is also provided to demonstrate how to iteratively run WinBUGS inside SAS for Monte Carlo simulation studies. The SAS codes used in this study are easily extended to accommodate many other models with only slight modification. This interface can be of practical benefit in many aspects of Bayesian methods because it allows the SAS users to benefit from the implementation of Bayesian estimation and it also allows the WinBUGS user to benefit from the data processing routines available in SAS.

Correspondence should be addressed to Zhiyong Zhang, Department of Psychology, University of Notre Dame, Notre Dame, IN 46556. E-mail: zzhang4@nd.edu

Bayesian methods have received more and more attention in social and behavioral researches (e.g., Edwards, Lindaman, & Savage, 1963; M. D. Lee, 2004; Myung & Pitt, 1997; Seltzer & Choi, 2003) and these models have been successfully applied to item response models (e.g., Chang, 1996; Fox & Glas, 2001), factor analytic models (e.g., Bartholomew, 1981; S. Lee, 1981), structural equation models (e.g., Congdon, 2003; Scheines, Hoijtink, & Boomsma, 1999), genetic models (e.g., Eaves & Erkanli, 2003), growth curve models (e.g., Zhang, Hamagami, Wang, Grimm, & Nesselroade, 2007), and multilevel models (e.g., Seltzer, Wong, & Bryk, 1996). In a recent debate by Trafimow (2003, 2005) and M. D. Lee and Wagenmakers (2005), the advantages and disadvantages of Bayesian methods were discussed and a promising future of Bayesian applications has been suggested. Based on the review of applications of Bayesian methods in social and behavioral research, Rupp, Dey, and Zumbo (2004) further concluded that both applied and theoretical communities could not afford to miss the opportunities opened up by Bayesian methods.

A drawback to the implementation of Bayesian analysis and estimation is the programming and computation demands. However, with the development of the computation capacity, the cost of computation is acceptable given its benefits. Furthermore, the emergence of the free available WinBUGS software (Spiegelhalter, Thomas, Best, & Lunn, 2003) has made programming much easier than before. WinBUGS is accepted as the most widely used and convenient tool for estimating both simple and complex Bayesian models (Congdon, 2001, 2003; Cowles, 2004).

A complete WinBUGS program consists of three parts: (a) model specification, (b) data input, and (c) starting values. Users first need to learn how to specify a model in WinBUGS syntax and these specifications vary across different models. In this article, we illustrate how to specify three models: a multiple regression model, a growth curve model, and a confirmatory factor model. Because the contributed WinBUGS example programs for many models are available freely on the WinBUGS development Web site, we focus on helping readers understand WinBUGS codes and customize the codes for their own empirical data analysis.

The data format in the data input and starting values parts of WinBUGS is very similar to Splus/R data format (Spiegelhalter et al., 2003). Researchers who are not familiar with the Splus/R data format may find it difficult to transform data to be compatible with WinBUGS. Fortunately, there are some free programs or macros that can transform different formats of data to WinBUGS format, such as an R function R2WinBUGS by Sturtz and Ligges, a set of SAS macros by Sparapani, an Excel macro xl2bugs by Misra, and a standalone program BAUW by Zhang and Wang (see Appendix for Internet links).

Although WinBUGS can be used as menu-driving software, we present a batch procedure to call and run WinBUGS inside a SAS script. This pro-

cedure has several advantages over other methods. First, it is easy for researchers who are already familiar with SAS to run WinBUGS. Second, it permits a researcher to use SAS procedures before and after Bayesian modeling. It is easy to describe and transform data first in SAS, run the WinBUGS program for Bayesian estimation, and then save and plot results using SAS procedures. Third, although WinBUGS provides a “menu” to run the program, the batch processing approach used here decreases the probability of mistakes and allows users to easily repeat similar analyses. This procedure is especially useful and convenient for analysis requiring repetition, such as in Monte Carlo simulation studies. This procedure is demonstrated in the following sections.

A COMPLETE PROCEDURE TO RUN WINBUGS INSIDE SAS

To run WinBUGS inside SAS, the procedure portrayed in Figure 1 can be followed. In the next sections, we describe this procedure using three examples: a multiple regression model, a linear growth curve model, and a confirmatory




FIGURE 1 The flow chart to run WinBUGS inside SAS.

factor analysis (CFA) model. The first example aims to demonstrate how to apply the whole procedure in Figure 1, including how to specify a model, transform data, create starting values, and run WinBUGS inside SAS to implement Bayesian analysis. The second example is based on the linear growth curve model (e.g., McArdle & Nesselroade, 2003; Meredith & Tisak, 1990) and aims to illustrate how to specify a more complex model. The third example aims to show how to iteratively run WinBUGS inside SAS for the Monte Carlo simulation study.

Example 1: The Multiple Regression Model

For the purpose of demonstration, we use a multiple regression model with two predictors. The model can be written as

$$y[i] = b_0 + b_1 * x_1[i] + b_2 * x_2[i] + e[i], \quad i = 1, \dots, N.$$

In its probability form, this model can be expressed as

$$\begin{aligned} y[i]|x_1[i], \quad x_2[i] &\sim N(\mu[i], \sigma_e^2) \\ \mu[i] &= b_0 + b_1 * x_1[i] + b_2 * x_2[i], \end{aligned} \quad (1)$$

where σ_e^2 is the residual or measurement error variance, b_0 is the intercept, and b_1 and b_2 are regression coefficients. A data set with the sample size $N = 1,000$ was generated from this model with the population parameter values set as $b_0 = 1$, $b_1 = 2$, $b_2 = 3$, and $\sigma_e^2 = 4$. The SAS codes for data generation are given in Code 1. The complete codes for running WinBUGS inside SAS to fit this multiple regression model are provided in Code 2 through Code 8.

Step 1: Install SAS and WinBUGS. The first step is to install SAS and WinBUGS. Because SAS is widely used, we assume that SAS has been installed and only focus on the installation of WinBUGS. WinBUGS is free software and can be downloaded from its Web site (see Appendix). The download is an executable file and it can be set up as a typical Windows program (double-click and follow screen instructions). WinBUGS requires a key for its unrestricted use. The key is sent by e-mail once the user completes a registration form on the WinBUGS Web site.

Step 2: Set up SAS environment for WinBUGS. A set of free SAS macros written by Sparapani (see Appendix) can be used to transform data from SAS data format to WinBUGS data format. The macros can be downloaded from their Web site (see Appendix A). A slightly modified version was used

```

DATA Sim_Reg;
b0=1; b1=2; b2=3; sig_e=2; seed=20060118; N = 1000;
DO _N_ = 1 TO N;
x1=RANNOR(seed);
    x2=RANNOR(seed);
    e=RANNOR(seed);
    y = b0+b1*x1+b2*x2+sig_e*e;
KEEP y x1 x2;
OUTPUT;
END;
RUN;

```

CODE 1 Data generation.

```

DATA model;
INPUT model $80.;
CARDS;/*start the model*/
model{
#Model specification
for (i in 1:N) {
y[i]~dnorm(muy[i], Inv_sig2_e)
    muy[i]<-b0+b1*x1[i]+b2*x2[i]
}
#priors
b0~dnorm(0, 1.0E-6)
b1~dnorm(0, 1.0E-6)
b2~dnorm(0, 1.0E-6)
Inv_sig2_e~dgamma(1.0E-3, 1.0E-3)
#parameter transformation
Sig2_e<-1/Inv_sig2_e
}
;
RUN;
DATA _NULL_;
SET model;
FILE "C:\SASWinBUGS\RegModel.txt";
PUT model;
RUN;

```

CODE 2 Model specification.

```

%_lexport(data=Sim_Reg, file='C:\SASWinBUGS\RegData.txt,'
var=y x1 x2);

```

CODE 3 Data transformation.

```
DATA _NULL_;
FILE "C:\SASWinBUGS\RegInit.txt";
PUT "list(b0=0, b1=0, b2=0, Inv_sig2_e=1)";
RUN;
```

CODE 4 Starting values specification.

```
DATA _NULL_;
FILE "C:\program files\WinBUGS14\RegBatch.txt";
PUT // @@
#1 "display('log')"
#2 "check('C:/SASWinBUGS/RegModel.txt')"
#3 "data('C:/SASWinBUGS/RegData.txt')"
#4 "compile(1)"
#5 "inits(1, 'C:/SASWinBUGS/RegInit.txt')"
#6 "gen.inits()"
#7 "update(2000)"
#8 "set(b0)"
#9 "set(b1)"
#10 "set(b2)"
#11 "set(Sig2_e)"
#12 "dic.set()"
#13 "update(5000)"
#14 "dic.stats()"
#15 "coda(*, 'C:/SASWinBUGS/output')"
#16 "save('C:/SASWinBUGS/bugslog.txt')"
#17 "quit()"
;
RUN;
```

CODE 5 Batch scripts to run WinBUGS.

```
DATA _NULL_;
FILE "C:\SASWinBUGS\runreg.bat";
PUT 'CD C:\program files\WinBUGS14';
PUT 'WinBUGS14.exe /PAR RegBatch.txt';
PUT 'EXIT';
RUN;
DATA _NULL_;
X "C:\SASWinBUGS\runreg.bat";
RUN; QUIT;
```

CODE 6 Run WinBUGS in SAS X window.

```

DATA log;
INFILE "C:\SASWinBUGS\bugslog.txt" TRUNCOVER;
INPUT log $80.;
log=translate(log, " ", "09"x);
RUN;
PROC PRINT DATA=log;
RUN;

```

CODE 7 View log file and DIC and debug errors.

```

%coda2sas(out=coda, infile='C:\SASWinBUGS\outputIndex.txt,'
chain='C:\SASWinBUGS\output1.txt,' stats=1);
QUIT;

```

CODE 8 Statistical inference.

in this article, which is also freely available. The following instructions can be followed to set up the macros.

1. Put the macros into a folder. To avoid unintentionally deleting these macros, we suggest putting the macros into the folder C:\Program Files\SAS\bugs, where bugs is a new folder that needs to be created first.
2. Modify the SAS configuration file, sasv8.cfg for SAS 8.x or sasv9.cfg for SAS 9.x. Open the file in Notepad and add the following two lines at the end and save it:

```

-insert sasautos 'C:\Program Files\SAS\bugs'
-insert sasautos '!SASROOT\core\sasmacro'

```

Step 3: Express the model in WinBUGS language. To implement Bayesian analysis in WinBUGS, we first need to express the models using WinBUGS syntax. For a typical WinBUGS program, the model specification part must include two subparts: the expression of the model and the choice of prior distributions (Spiegelhalter et al., 2003). For the regression model, the codes are given in Code 2.

All WinBUGS programs start with a keyword `model` and the whole model specification part needs to be put within a pair of brackets `{ }`. The first section of the model specification part can be viewed as the direct translation of the probability form of the model. For the regression model in Equation 2, WinBUGS codes for the i th individual were

$$y[i] \sim \text{dnorm}(muy[i], \text{Inv_sig2_e}), \text{ and}$$

$$muy[i] <- b0 + b1 * x1[i] + b2 * x2[i].$$

The first line indicates that $y[i]$ had (\sim) a normal distribution (`dnorm`) with two arguments: mean `muy[i]` and precision `Inv_sig2_e`. The precision in the second argument is the reciprocal of the variance ($1/\sigma_e^2$). The mean was equal to (`<-`) the combination of the two predictors `x1` and `x2` with the regression coefficients `b1` and `b2`. Because we had $N = 1,000$ individuals, we used a `for (i in 1:N){ . . . }` loop to repeat this specification for each individual. `for` is the keyword for a loop. `i in 1:N` in the parentheses means replacing `i` using `1, 2, . . . , N` and then implement everything in the brackets `{ }` following `for (i in 1:N)`.

In the second section of the model specification part, we need to choose a prior distribution for each parameter in the model. For this regression model, there were four parameters, `b0`, `b1`, `b2`, and `Inv_sig2_e`. For the regression intercept and regression coefficients, the normal distribution priors with mean 0 and precision `1.0E-6` were specified as `bk~dnorm(0, 1.0E-6)` with $k = 0, 1, 2$, which is a widely used noninformative prior (Congdon, 2001, 2003). For the precision parameter, a Gamma distribution (`dgamma`) prior with shape and scale parameter = `1.0E-3` was used `Inv_sig2_e~dgamma(1.0E-3, 1.0E-3)`, which is also a widely used noninformative prior for the variance parameter (Congdon, 2001, 2003). Finally, we transformed the precision back to the variance.

Running the codes in Code 2 saved the WinBUGS model specification codes for the multiple regression model into a file called `RegModel.txt`.

Step 4: Configure a SAS program to run WinBUGS. In this step, we configured a SAS program to create WinBUGS compatible data, construct a starting value file, run WinBUGS, and analyze the Markov chain Monte Carlo (MCMC) data generated by WinBUGS. The following five steps can be performed in SAS.

Step 4-1: Create a WinBUGS data file. By using the SAS macros in Step 2, a SAS data set can be converted to the WinBUGS format. WinBUGS uses a keyword `list(. . .)` to organize the data. The data in the parentheses can be a scalar, a vector, or an array. For a scalar, the format is `ScalarName=data`. For example, $N=1000$ and $T=5$. For a vector, the format is `VectorName=c(data1, data2, . . . , dataN)`. The keyword `c` combines the values separated by the comma in the parentheses into a vector. For example, `Mu=c(0,0)` is a vector with two elements. For an array, the format is `ArrayName=structure(.Data = c(data1, data2, . . .), .Dim = c(nrow, ncol, . . .))`. WinBUGS reads data in `.Data = c(data1, data2, . . .)` into an array by filling the right-most index for dimensions in `.Dim = c(nrow, ncol, . . .)` first. For example, for a two-dimension (3×2) array,

```
y = structure(.Data = c(1, 2, 3, 4, 5, 6),
```

$$.Dim = c(3, 2)) = \begin{pmatrix} 1 & 2 \\ 3 & 4 \\ 5 & 6 \end{pmatrix}.$$

The SAS macros set up in Step 2 can convert SAS data into a list with vector (`_lexport`) or array (`_sexport`). Using the codes in Code 3, a SAS data set was converted to WinBUGS vector data and saved into a file. In this example, the macro `_lexport` was used. `data=` defines the SAS data set to be used. Here it was a SAS data set called `Sim_Reg`. `file=` defines the file to save the data in WinBUGS data format. `var=` defines the variables to transform. In the generated data file, three vectors, `y`, `x1`, and `x2`, along with a scalar representing the sample size (`N=1000`) were saved.

Step 4–2: Create a starting value file. For each parameter in the model, a starting value needs to be either specified manually or generated by WinBUGS. Usually, for each canonical parameter, we give it a starting value manually and let WinBUGS generate the others. The format of the starting values has the same format as the data. The same way to transform data can be used to create starting values. Usually, the starting values are relatively easier to handle and can be put together directly. The SAS codes in Code 4 created a starting value file for the regression model. All the starting values were put together in a list and saved in the file `RegInit.txt`.

Step 4–3: Create a script file to run WinBUGS. In this step, we used the batch mode to run WinBUGS. Code 5 provides the SAS codes for the regression model.

Line 1 opened a log window to trace the history and errors of the implementation process. Line 2 checked whether the syntax in the file `RegModel.txt` was correct. Line 3 read in the data, Line 4 compiled the model, Line 5 initialized the parameters using the data in the starting value file, and Line 6 generated the starting values for the parameters that were not specified in the starting value file. Line 7 generated 2,000 data points for each parameter but these data points (called burn-in data points) were discarded to ensure convergence. In Lines 8 through 11 the parameters were specified to be estimated. The parameters need to be specified because the sampled data points for unspecified parameters will not be saved. Line 12 was used to monitor the deviance information criterion (DIC) (Spiegelhalter, Best, Carlin, & Linde, 2002) that can be used as a fit statistic to compare models. Line 13 generated the other 5,000 data points that

were saved to be analyzed in SAS for statistical inference. Line 14 wrote the DIC into the log file. Line 15 saved the generated data points in Line 13 into two files. The first file was the index file `outputIndex.txt` that included the index for each estimated parameter defined in Lines 8 through 11. The second file was the data file `output1.txt` that saved the data points generated. These files were also called convergence diagnostic and output analysis (CODA) files. Line 16 saved the log file where the DIC and error information can be found. Line 17 was used to quit the WinBUGS program when finished.

Three comments are worth emphasizing here. First, this script file must be saved where WinBUGS is installed, usually `C:\program files\WinBUGS14\`. Second, one can change the folder where the model file, data file, starting value file, coda file, and log file are saved. However, the slash (/) instead of the usual backslash (\) needs to be used in the path. Third, WinBUGS is sensitive to lowercase and uppercase letters.

Step 4–4: Run WinBUGS and debug errors. To run WinBUGS, we first created a `run.bat` file using the codes in the first paragraph of Code 6. We then ran WinBUGS in the X window using the codes in the second paragraph. After implementing the first two paragraphs of the codes, a DOS window opened and the WinBUGS program implemented the procedure specified in Step 4–3. After finishing, the WinBUGS program exited and the SAS window returned. The first thing to check then is the log file. The codes in Code 7 can read the log file and print its content in the output window of SAS. Any errors in running WinBUGS can be targeted by the information provided in the log file.

Step 4–5: Read CODA files into SAS and implement statistical inferences. The CODA file generated by WinBUGS can be read into SAS using the macro `coda2sas` with the codes in Code 8. The first argument “`out=`” specified the name of a SAS data set to save the generated data points. Here a data set called “`coda`” was created. “`infile=`” specified the index file and “`chain=`” specified the data file saved in Step 4–3. By specifying “`stats=1,`” the macro `coda2sas` generated the history plot and the histogram with the overlaid kernel density and calculated the descriptive statistics for each parameter. If more analyses are needed, one can work on the data set “`coda.`” For the regression model, the history and histogram plots for the regression model are given in Figure 2. From the history plots, the generated sequences for all parameters converged through the “eyeball” check. The density plots and descriptive statistics for the four model parameters are given in Figure 2 and Table 1, respectively. From Table 1, the estimated parameters were very close to the population values used to generate the data.




FIGURE 2 History plots and histogram plots of parameters from the regression model.

TABLE 1
The Parameter Estimates for the Regression Model

	<i>True</i>	<i>Estimate</i>	<i>SE</i>	<i>CI</i>
b0	1	0.99	0.064	0.87, 1.12
b1	2	2.04	0.065	1.92, 2.17
b2	3	3.01	0.064	2.89, 3.14
sig2_e	4	4.06	0.184	3.71, 4.45

Note. SE = standard error; CI = confidence interval.

Example 2: The Linear Growth Curve Model

We have shown how to use the procedure in Figure 1 to estimate a multiple regression model. In this example, we present a linear growth curve model to demonstrate how to specify a more complex model using WinBUGS syntax. The linear growth curve model (e.g., McArdle & Nesselroade, 2003) can be written as

$$y[i, t] = L[i] + t \times S[i] + e[i, t]$$

$$L[i] = \mu_L + v_L[i]$$

$$S[i] = \mu_S + v_S[i]$$

$$i = 1, \dots, N; \quad t = 1, \dots, T,$$

where $y[i, t]$ represents the observed score for the i th individual at occasion t , $L[i]$ represents the level and $S[i]$ represents the slope for the i th individual, $e[i, t]$ represents the measurement error, μ_L and μ_S are the average level and slope of N individuals, and $v_L[i]$ and $v_S[i]$ are the individual deviances for the initial level and slope from the average level and slope for i th individual.

Using probability density function, this model can be expressed as

$$y[i, t] | L[i], S[i] \sim N(\mu[i, t], \sigma_e^2)$$

$$\mu[i, t] = L[i] + t \times S[i] \quad (2)$$

$$\begin{pmatrix} L[i] \\ S[i] \end{pmatrix} \sim MN \left(\begin{bmatrix} \mu_L \\ \mu_S \end{bmatrix}, \begin{bmatrix} \sigma_L^2 & \sigma_{LS} \\ \sigma_{LS} & \sigma_S^2 \end{bmatrix} \right)$$

$$i = 1, \dots, N; \quad t = 1, \dots, T,$$

where N represents the univariate normal distribution, σ_e^2 represents the variance of measurement errors, MN represents the multivariate normal distribution, σ_L^2 and σ_s^2 represent the variances of the level and slope respectively, and σ_{LS} is the covariance between the level and slope. Equation 2 indicates that the level and slope have a bivariate normal distribution and the observed variable has a univariate normal distribution with the mean expressed as the combination of the level and slope.

Based on this linear growth curve model, we simulated a data set with $N = 1,000$ participants, and $T = 5$ occasions using SAS with the population parameter values $\mu_L = 10$, $\mu_S = 5$, $\sigma_e^2 = 1$, $\sigma_S^2 = 4$, $\sigma_s^2 = 1$, and $\sigma_{LS} = 1$ or $\rho_{LS} = .5$. Because the maximum likelihood estimation (MLE) is commonly applied to obtain parameter estimates for the linear growth curve model (e.g., Demidenko, 2004; Laird & Ware, 1982), we briefly compare the results from Bayesian estimation (BE) with those from MLE.

For the linear growth curve model, the model specification part is given in Code 9. Because the level and slope are bivariate normally distributed, we need to specify a bivariate normal distribution for them, which is `dmnorm` in WinBUGS. The bivariate normal distribution has two arguments: mean vector and covariance matrix. In WinBUGS, the second argument for `dmnorm` is the precision, which is the inverse of the covariance matrix. In WinBUGS, this bivariate distribution was expressed as

$$LS[i, 1 : 2] \sim \text{dmnorm}(\text{Mu}[1 : 2], \text{Inv_cov}[1 : 2, 1 : 2]),$$

where $LS[i, 1:2]$ is a 2×1 vector with two elements $LS[i, 1]$ and $LS[i, 2]$; $LS[i, 1]$ is the level and $LS[i, 2]$ is the slope for individual i ; $\text{Mu}[1:2]$ is a 2×1 mean vector; and $\text{Inv_cov}[1:2, 1:2]$ is the inverse of the covariance matrix of the initial level and slope. Because we had $N = 1,000$ individuals, we used a `for (i in 1:N){ . . . }` loop to repeat this specification for each individual. The observed variable y had a univariate normal distribution that was expressed as

$$y[i, t] \sim \text{dnorm}(\text{MuY}[i, t], \text{Inv_sig2_e}).$$

Because each individual had an observation from occasion 1 to $T = 5$, we used a second loop `for (t in 1:T)` nested in the first one to represent this.

For the linear growth model, there were six parameters, $\text{Mu}[1,2]$, $\text{Inv_cov}[1:2, 1:2]$, and Inv_sig2_e . We gave the mean vector $\text{Mu}[1,2]$ a bivariate normal distribution prior. For the precision matrix ($\text{Inv_cov}[1:2, 1:2]$), a Wishart distribution prior was used because it is the multivariate generalization of the Gamma distribution. For the precision of y , a Gamma distribution prior was used.

```

DATA model;
INPUT model $80.;
CARDS; /*Start of the model scripts*/
  model
  {#Model
  for (i in 1:N){
  LS[i,1:2]~dmnorm(Mu[1:2], Inv_cov[1:2,1:2])
  for (t in 1:T){
  y[i,t]~dnorm(MuY[i,t], Inv_sig2_e)
  MuY[i,t]<-LS[i,1]+LS[i,2]*t
  }
  }
  #Prior
  Mu[1:2]~dmnorm(Mu0[1:2], Inv_cov0[1:2,1:2])
  Mu0[1]<-0
  Mu0[2]<-0
  Inv_cov0[1,1]<-1.0E-6
  Inv_cov0[2,2]<-1.0E-6
  Inv_cov0[2,1]<-Inv_cov0[1,2]
  Inv_cov0[1,2]<-0
  Inv_cov[1:2,1:2]~dwish(R[1:2,1:2], 2)
  R[1,1]<-1
  R[2,2]<-1
  R[2,1]<-R[1,2]
  R[1,2]<-0
  Inv_sig2_e~dgamma(.001,.001)
  #Transform of the parameters
  MuL<-Mu[1]
  MuS<-Mu[2]
  Cov[1:2,1:2]<-inverse(Inv_cov[1:2,1:2])
  Sig2_L<-Cov[1,1]
  Sig2_S<-Cov[2,2]
  rho<-Cov[1,2]/sqrt(Cov[1,1]*Cov[2,2])
  Sig2_e<-1/Inv_sig2_e
  }
  ;
/*end of the model scripts*/
RUN;
DATA _NULL_;
SET model;
FILE 'C:\SASWinBUGS\GrowthModel.txt';
PUT model;
RUN;

```

CODE 9 Model specification of the linear growth model.

```

%_sexport(data=Sim_LinGM,
          file='C:\SASWinBUGS\GrowthData.txt,'
          var=y1-y5);

DATA _NULL_;
FILE "C:\SASWinBUGS\InitValues.txt";
PUT "list(Mu=c(0,0), Inv_cov= structure(.Data =
      c(1,0,0,1),.Dim=c(2,2)), Inv_sig2_e=1) ";

```

CODE 10 Data transformation and starting values for the linear growth curve model.

Running the codes in Code 9 saved the WinBUGS scripts for the growth curve model into a file called GrowthModel.txt.

In this example, we used the $N \times T$ ($1,000 \times 5$) array data $y[i, t]$. Using the scripts in Code 10, the SAS data set `Sim_LinGM` was converted to WinBUGS array data using macro `_sexport` and saved into a file called `GrowthData.txt`. All the starting values were put together in a `list` and saved in the file `InitValues.txt`. Note that this set of starting values included all three types of data.

The batch scripts and the .bat file for this linear growth curve model were very similar to those for the regression model and we did not repeat them here. The complete SAS codes for this model are available by request. After running the complete SAS codes, all parameter estimates from WinBUGS along with those from SAS MIXED are summarized in Table 2. From Table 2, the parameter estimates from WinBUGS were very close to the population values. Furthermore, the parameter estimates from WinBUGS and SAS MIXED were nearly identical, which demonstrates that Bayesian method estimation provides the same level of accuracy as MLE when noninformative priors are used.

TABLE 2
Parameter Estimates for the Linear Growth Model

	True Value	WinBUGS		SAS MIXED	
		Estimate	SE	Estimate	SE
μ_L	10	10.12	0.071	10.12	0.071
μ_S	5	5.01	0.033	5.01	0.034
σ_L^2	4	3.96	0.227	3.96	0.226
σ_S^2	1	1.04	0.051	1.04	0.051
σ_e^2	1	0.97	0.025	0.97	0.025
ρ_{LS}	0.5	0.45	0.034	0.44	0.038

Note. SE = standard error.

Example 3: Monte Carlo Simulation of a Confirmatory Factor Model

Bayesian methods have been mainly used as an alternative to MLE or to estimate complex models that usually cannot be easily estimated with MLE. Simulation studies are necessary when evaluating new or complex models. WinBUGS is not very flexible for simulation studies because it can only run a single model or a single data set at one time. However, SAS can be used to iteratively implement the simulation procedure. To demonstrate how to use SAS to iteratively run WinBUGS, we use a confirmatory factor model with one latent factor and four observed variables. A path diagram with the population parameter values is plotted in Figure 3. We generated 100 sets of data from the population models and parameter estimates were obtained for each data set using WinBUGS. We compared the mean of the parameter estimates from all 100 sets of data with the population values.

For the simulation study, the model specification, starting values, and script file to run WinBUGS are the same for each data set. The codes in Code 11 can be used to set up those for the confirmatory factor model. In this example,




FIGURE 3 Path diagram for the population confirmatory factor model.

```

TITLE "Model specification for the CFA";
FILENAME model "C:\SASWinBUGS\cfamodel.txt";
DATA model;
  INPUT model $80.;
  CARDS;/*start the model*/
  model{
    for (i in 1:N){
      for (t in 1:T){
        y[i,t]~dnorm(muy[i,t],Inv_sig2[t])
        muy[i,t]<-fload[t]*fscore[i]
      }
      fscore[i]~dnorm(0, 1)
    }
    for (t in 1:T){
      fload[t]~dnorm(0, 1.0E-6)
      Inv_sig2[t]~dgamma(0.001, .001)
      Para[t]<-fload[t]
      Para[t+4]<-1/Inv_sig2[t]
    }
  }
  ;
RUN;
DATA _NULL_;
SET model;
FILE model;
PUT model;
RUN;
TITLE "Starting values for CFA";
DATA _NULL_;
  FILE "C:\SASWinBUGS\cfaini.txt";
  PUT "list(fload=c(.5,.5,.5,.5), Inv_sig2=c(1,1,1,1))";
RUN;
TITLE "Batch scripts to run WinBUGS";
FILENAME runcfa 'c:\program files\winbugs14\runcfa.txt';
DATA _NULL_;
FILE runcfa;
PUT@1 "display('log')";
PUT@1 "check('C:/SASWinBUGS/cfamodel.txt')" ;
PUT@1 "data('C:/SASWinBUGS/cfadata.txt')";
PUT@1 "compile(1)";
PUT@1 "inits(1, 'C:/SASWinBUGS/cfaini.txt')";
PUT@1 "gen.inits()";
PUT@1 "update(2000)";
PUT@1 "set(Para)";
PUT@1 "update(3000)";
PUT@1 "stats(*)";
PUT@1 "save('C:/SASWinBUGS/cfalog.txt')";

```

CODE 11 Common scripts for the CFA simulation (*continued on next page*).

```

PUT@1 "quit()";
RUN;
DATA _NULL_;
  FILE "C:\SASWinBUGS\runcfa.bat";
  PUT 'C:\program files\WinBUGS14\WinBUGS14.exe' /PAR
      runcfa.txt';
  PUT 'exit';
RUN;

```

CODE 11 (Continued).

a new WinBUGS command “stat()” was used, which calculated the summary statistics for each parameter inside WinBUGS.

For the Monte Carlo simulation study, each data set generated from the population model was different and the parameter estimates from each data set generally were likely to be somewhat different. Thus, we need to generate multiple data sets and estimate the parameters for each data set iteratively. To do this, we used a macro that can be called iteratively. Each time this macro was called, it generated a data set and obtained parameter estimates from the model. The macro `simcfa(n)` for the CFA is given in Code 12. In the first part, a data set was generated from the population model. Then this data set was saved into a file `cfadata.txt` in WinBUGS data format. In the next part, WinBUGS was run in an X window to implement the Bayesian analysis based on this generated data set. Finally, the log file was read into SAS to obtain the parameter estimates. Notice that we did not save the CODA files and calculate the parameter estimates in SAS. Instead, we read in the parameter estimates from the log file directly. In this case, we need to make sure the generated sequences converged. In this example, we first ran one set of data and found that the generated sequences for all parameters converged after 100 iterations. Although we can use 100 as the burn-in data points, we used 2,000 to ensure the convergence for all the other data sets.

To run the macro in Code 12 100 times to generate 100 data sets and obtain 100 sets of parameters, we configured another macro, `runsimcfa`, which is given in Code 13.

After running the macro in Code 13, the 100 sets of parameter estimates were printed in the SAS output window. Usually, for each parameter, we need to calculate three numbers: the mean and the standard deviation of each parameter estimate, and the mean of the associated standard errors (MSE) from the 100 sets of data. All of these can be calculated using the SAS codes in Code 14.

After running Code 11 through Code 14, we can obtain the results shown in Table 3. The means of the parameter estimates are very close to the population parameter values to generate the data. Furthermore, the standard deviations were the same as the MSE, indicating the estimated standard errors were consistent with the true standard errors.

```

%MACRO simcfa(n);
TITLE 'Generate the Data';
DATA Sim_CFA;
*setting the true parameter values;
fload=.8; sig2=.36;
* setting statistical parameters;
N = 200; seed = 20060802+&n; M=4;
* need to setup arrays so we can have more variables;
ARRAY y_score{4} y1-y4;
ARRAY e_score{4} y1-y4;
* generating raw data;
DO _N_ = 1 TO N;
* now the indicator variables ;
f_score=RANNOR(seed);
  DO t = 1 TO M;
    y_score{t} = fload*f_score +sqrt(sig2)*RANNOR(seed);
  END;
KEEP y1-y4;
OUTPUT;
END;
RUN;
/*Data*/
%_sexport(data=Sim_CFA, file='C:\SASWinBUGS\cfadata.txt,'
var=y1-y4);
/*Run WinBUGS*/
DATA _NULL_;
  X "C:\SASWinBUGS\runcfa.bat";
RUN;
QUIT;
/*Read in the log file to view the parameters*/
TITLE 'Simulation '&n;
DATA log;
  INFILE "C:\SASWinBUGS\cfalog.txt" TRUNCOVER ;
  INPUT log $80.;
  log=translate(log," ","09"x);
  IF (SUBSTR(log, 2, 4) ne 'Para') then delete;
RUN;
PROC PRINT DATA=log;
RUN;
%MEND;

```

CODE 12 The macro for data generation and model estimation.

DISCUSSION

WinBUGS is a powerful tool for implementing Bayesian analysis and estimating complex models (Rupp et al., 2004) and SAS is widely used statistical software in academic and research institutes. The combination of these tools will advance

```

%MACRO runsimcfa;
%LET n=1;
%DO %WHILE(&n <= 100);
%simcfa(&n);
%LET n=%EVAL(&n+1);
%END;
%MEND runsimcfa;
*run the macro
%runsimcfa;

```

CODE 13 The macro for running simulation iteratively.

the application of both Bayesian methods and sophisticated models in social and behavioral research. The whole procedure we presented and the SAS codes we provided can conveniently interface SAS and WinBUGS. This procedure is beneficial to many researchers, including advanced Bayesian users who already have rich experiences in Bayesian analysis and researchers who are familiar with SAS but have yet to discover the utility of Bayesian approach.

The procedure in Figure 1 was illustrated using a multiple regression model, a linear growth curve model, and a confirmatory factor model. In the first example, we demonstrated how to apply the proposed interface between SAS and WinBUGS using a multiple regression model. The second example showed how to specify a more complex model, and the last example focused on the iterative use of WinBUGS for Monte Carlo simulation studies. All three examples can be replicated and modified to accommodate new models.

TABLE 3
Results From the Confirmatory Factor Analysis

<i>Parameters</i>	<i>True</i>	<i>M</i>	<i>SD</i>	<i>MSE</i>
Factor loadings	0.8	0.81	0.06	0.06
	0.8	0.82	0.06	0.06
	0.8	0.80	0.06	0.06
	0.8	0.81	0.06	0.06
Uniqueness variances	0.36	0.35	0.04	0.05
	0.36	0.37	0.05	0.05
	0.36	0.38	0.05	0.05
	0.36	0.36	0.05	0.05

Note. *M* = the average value of the parameter estimates from 100 sets of simulated data; *SD* = standard deviation of the parameter estimates from 100 sets of simulated data; *MSE* = average standard errors of the parameter estimates from 100 sets of simulated data.

```

/*Save the output and log into files*/
DM OUTPUT 'FILE "C:\SASWinBUGS\allresults.txt"';
DM LOG 'FILE "C:\SASWinBUGS\allresults.log"';
TITLE "Analyze the Monte Carlo simulation results";
DATA temp;
  INFILE "C:\SASWinBUGS\allresults.txt" TRUNCOVER ;
  INPUT all $90.;
  IF (SUBSTR(all, 7, 4) NE 'Para') THEN DELETE;
  FILE "C:\SASWinBUGS\temp.txt";
  PUT all;
RUN;
DATA temp;
INFILE "C:\SASWinBUGS\temp.txt";
INPUT parid parname $ parest parsd MError p25 median p975 start sample;
id=int((_N_.-1)/8)+1;
parest=abs(parest);
RUN;
/*Parameter Estimates*/
PROC TRANSPOSE DATA=temp OUT=parest PREFIX=par;
BY id ;
ID parid;
VAR parest;
RUN;
/*Calculate the mean and s.d. of the parameters*/
PROC MEANS DATA=parest;
VAR par1-par8;
RUN;
/*SDs*/
PROC TRANSPOSE DATA=temp OUT=parsd PREFIX=sd;
BY id ;
ID parid;
VAR parsd;
RUN;
/*Calculate the mean of the s.e.*/
PROC MEANS DATA=parsd;
VAR sd1-sd8;
RUN;

```

CODE 14 Data process of the simulation results.

Two concerns about the Bayesian analysis include the computational time and programming intensity required. However, with the availability of powerful computing facilities, computing time is not an obstacle anymore. For example, the multiple regression model took about 10 sec to finish the estimation procedure on an outdated laptop (Celeron 1.7 MHz processor and 512 MB of RAM). The growth curve model with $N = 1,000$ and $T = 5$ took about 120 sec. For the confirmatory factor model example, it took only 30 min to finish the whole simulation study (with 100 sets of data). Furthermore, although

the different models need different model specification, the example WinBUGS codes for many models can be obtained freely. By simply replacing the model specification part in our example and making a few other minor changes, a new data set can be analyzed by a new model.

For the aim of illustration, we used three relatively simple models as examples. However, the same procedure allows and shows advantages when estimating more complex models that cannot be analyzed in SAS easily, such as the change point models (McArdle & Wang, 2008; Wang & McArdle, 2008), dynamic item response models (Ram et al., 2005), and categorical dynamic factor models (Zhang & Nesselroade, 2007). Complexity of models also made the difference in computation time less noticeable between Bayesian and MLE methods. However, the precision of the parameter estimates is even better for Bayesian methods.

To close the discussion, we would like to evaluate the proposed procedure based on our practical experience. First, this procedure is very useful for analyzing data using a similar model. For example, analyzing the cognitive data using the same linear growth curve model, we only need to import the data into SAS and run the exact same procedure, changing only the name of the data set. Second, this procedure is especially useful for simulation studies. It is well known that WinBUGS is not flexible for simulation studies because it can run only a single replication at one time. Our procedure can be viewed as a useful supplement to WinBUGS. This procedure has been proved useful in Bowels (2006), Zhang, Hamaker, and Nesselroade (2008), and Zhang and Nesselroade (2007) for different simulation studies.

REFERENCES

- Bartholomew, D. J. (1981). Posterior analysis of the factor model. *British Journal of Mathematical and Statistical Psychology*, *34*, 93–99.
- Bowels, R. P. (2006). *Item response models for intratask change to examine the impacts of proactive interference on the aging of working memory span*. Unpublished doctoral dissertation, Department of Psychology, University of Virginia, Charlottesville, VA.
- Congdon, P. (2001). *Bayesian statistical modeling*. New York: Wiley.
- Congdon, P. (2003). *Applied Bayesian modeling*. New York: Wiley.
- Chang, H.-H. (1996). The asymptotic posterior normality of the latent trait for polytomous IRT models. *Psychometrika*, *61*, 445–463.
- Cowles, M. K. (2004). Review of WinBUGS 1.4. *The American Statistician*, *58*, 330–336.
- Demidenko, E. (2004). *Mixed models: Theory and applications*. New York: Wiley.
- Eaves, L., & Erkanli, A. (2003). Markov chain Monte Carlo approaches to analysis of genetic and environmental components of human developmental change and G X E interaction. *Behavior Genetics*, *33*, 279–299.
- Edwards, W., Lindman, H., & Savage, L. J. (1963). Bayesian statistical inference for psychological research. *Psychological Review*, *70*, 193–242.

- Fox, J.-P., & Glas, C. A. W. (2001). Bayesian estimation of a multilevel IRT model using Gibbs sampling. *Psychometrika*, *66*, 271–288.
- Laird, N. M., & Ware, J. H. (1982). Random-effects models for longitudinal data. *Biometrics*, *38*, 963–974.
- Lee, M. D. (2004). A Bayesian analysis of retention functions. *Journal of Mathematical Psychology*, *48*, 310–321.
- Lee, M. D., & Wagenmakers, E. (2005). Bayesian statistical inference in psychology: Comment on Trafimow (2003). *Psychological Review*, *112*, 662–668.
- Lee, S. (1981). A Bayesian approach to confirmatory factor analysis. *Psychometrika*, *46*, 153–160.
- McArdle, J. J., & Nesselroade, J. R. (2003). Growth curve analysis in contemporary psychological research. In J. Schinka & W. Velicer (Eds.), *Comprehensive handbook of psychology: Research methods in psychology* (Vol. 2, pp. 447–480). New York: Wiley.
- McArdle, J. J., & Wang, L. (2008). Modeling age-based turning points in longitudinal Life-span growth curves of cognition. In P. Cohen (Ed.), *Turning points research* (pp. 105–128). Mahwah, NJ: Lawrence Erlbaum Associates, Inc.
- Meredith, W., & Tisak, J. (1990). Latent curve analysis. *Psychometrika*, *55*, 107–122.
- Myung, I. J., & Pitt, M. A. (1997). Applying Occam's razor in modeling cognition: A Bayesian approach. *Psychonomic Bulletin & Review*, *4*(1), 79–95.
- Ram, N., Chow, S., Bowles, R. P., Wang, L., Grimm, K., Fujita, F., et al. (2005). Examining interindividual differences in cyclicity of pleasant and unpleasant affect using spectral analysis and item response modeling. *Psychometrika*, *70*, 773–790.
- Rupp, A. A., Dey, D. K., & Zumbo, B. D. (2004). To Bayes or not to Bayes, from whether to when: Applications of Bayesian methodology to modeling. *Structural Equation Modeling*, *11*, 424–451.
- Scheines, R., Hoijtink, H., & Boomsma, A. (1999). Bayesian estimation and testing of structural equation models. *Psychometrika*, *64*, 37–52.
- Seltzer, M., & Choi, K. (2003). Sensitivity analysis for hierarchical models: Downweighting and identifying extreme cases using the t distribution. In S. P. Reise & N. Duan (Eds.), *Multilevel modeling: Methodological advances, issues, and applications* (pp. 25–52). Mahwah, NJ: Lawrence Erlbaum Associates, Inc.
- Seltzer, M. H., Wong, W. H., & Bryk, A. S. (1996). Bayesian analysis in applications of hierarchical models: Issues and methods. *Journal of Educational and Behavioral Statistics*, *21*, 131–167.
- Spiegelhalter, D. J., Best, N. G., Carlin, B. P., & Linde, A. v. d. (2002). Bayesian measures of model complexity and fit. *Journal of the Royal Statistical Society: Series B, Statistical Methodology*, *64*, 583–639.
- Spiegelhalter, D. J., Thomas, A., Best, N., & Lunn, D. (2003). *WinBUGS manual version 1.4*. Cambridge, UK: MRC Biostatistics Unit, Institute of Public Health.
- Trafimow, D. (2003). Hypothesis testing and theory evaluation at the boundaries: Surprising insights from Bayes's theorem. *Psychological Review*, *110*, 526–535.
- Trafimow, D. (2005). The ubiquitous Laplacian assumption: Reply to Lee and Wagenmakers (2005). *Psychological Review*, *112*, 669–674.
- Wang, L., & McArdle, J. J. (2008). Estimating unknown change points by using Bayesian methods. *Structural Equation Modeling*, *15*, 52–74.
- Zhang, Z., Hamagami, F., Wang, L., Grimm, K. J., & Nesselroade, J. R. (2007). Bayesian analysis of longitudinal data using growth curve models. *International Journal of Behavioral Development*, *31*, 374–383.
- Zhang, Z., Hamaker, E. L., & Nesselroade, J. R. (2008). Comparisons of four methods for estimating dynamic factor models. *Structural Equation Modeling*, *15*(3), 377–402.
- Zhang, Z., & Nesselroade, J. R. (2007). Bayesian estimation of categorical dynamic factor models. *Multivariate Behavioral Research*, *42*, 729–756.

APPENDIX
LIST OF PROGRAMS AND MACROS

1. SAS: <http://www.sas.com>
2. WinBUGS: <http://www.mrc-bsu.cam.ac.uk/bugs/welcome.shtml>
3. R2WinBUGS by Sibylle Sturtz and Uwe Ligges: <http://cran.r-project.org/src/contrib/Descriptions/R2WinBUGS.html>
4. SAS Macros by Rodney Sparapani: <http://www.mcw.edu/pcor/bugs/>
5. xl2bugs by Sanjog Misra: <http://smisra.simon.rochester.edu/software.htm>
6. BAUW by Zhiyong Zhang and Lijuan Wang: <http://bauw.psychstat.org>
7. Modified version of Rodney Sparapani's SAS Macros: <http://bauw.psychstat.org>